

# FUNCTIONS and PROCEDURES SYSTEM DESIGN

*It is time to integrate several circuits into a bigger system*

**VHDL**  
**VHSIC HARDWARE**  
**DESCRIPTION LANGUAGE**

# FUNCTIONS and PROCEDURES

## FUNCTION Definition

- FUNCTIONS and PROCEDURES are:
  - known as *subprograms*.
  - a piece of **sequential** VHDL code
  - Very similar to PROCESS - The use of **IF**, **CASE** and **LOOP** are allowed (WAIT is not allowed).
  - PROCESS is thought for immediate use. However, **FUNCTIONS and PROCEDURES are thought for LIBRARY** allocation. That is their main difference.
  - The later does not means that FUNCTIONS and PROCEDURES can not be also used in main code,
- A FUNCTION has a **BODY** and a **CALL**, with the following syntax:

### FUNCTION BODY:

```
FUNCTION function_name [<parameter list>] RETURN data_type IS
    [declarations]
BEGIN
    (sequential statements)
END function_name;
```

In this example a, b are constants (ommission of CONSTANT is allowed), c is a SIGNAL. Data type of a and b is INTEGER and c is STD\_LOGOC\_VECTOR.

- <parameter list>** can be CONSTANT or SIGNAL (can not be a VARIABLE)
- data\_type** can be any (STD\_LOGIC, INTEGER, etc) but no range specification is allowed. (dont use: RANGE, TO , DOWNT0, etc)
- Only one return value is allowed.

```
FUNCTION f1 (a, b: INTEGER; SIGNAL c: STD_LOGIC_VECTOR)
    RETURN BOOLEAN IS
BEGIN
    (sequential statements)
END f1;
```

# FUNCTIONS and PROCEDURES

## FUNCTION Definition

- FUNCTIONS and PROCEDURES are:
  - known as *subprograms*.
  - a piece of **sequential** VHDL code
  - Very similar to PROCESS - The use of **IF**, **CASE** and **LOOP** are allowed (WAIT is not allowed).
  - PROCESS is thought for immediate use. However, **FUNCTIONS and PROCEDURES are thought for LIBRARY** allocation. That is their main difference.
  - The later does not means that FUNCTIONS and PROCESS can not be also used in main code,
- A FUNCTION has a **BODY** and a **CALL**, with the following syntax:

### FUNCTION CALL:

```
x <= conv_integer(a);      -- converts a to an integer
                           -- (expression appears by itself)
y <= maximum(a, b);        -- returns the largest of a and b
                           -- (expression appears by itself)
IF x > maximum(a, b) ...    -- compares x to the largest of a, b
                           -- (expression associated to a
                           -- statement)
```

- A **FUNCTION** is **called** as part of an expression associated to a **concurrent or sequential statement**.

## FUNCTIONS and PROCEDURES

### FUNCTION Example


```
----- Function body: -----  
FUNCTION conv_integer (SIGNAL vector: STD_LOGIC_VECTOR)  
    RETURN INTEGER IS  
    VARIABLE result: INTEGER RANGE 0 TO 2**vector'LENGTH-1;  
BEGIN  
    IF (vector(vector'HIGH)='1') THEN result:=1;  
    ELSE result:=0;  
    END IF;  
    FOR i IN (vector'HIGH-1) DOWNT0 (vector'LOW) LOOP  
        result:=result*2;  
        IF(vector(i)='1') THEN result:=result+1;  
        END IF;  
    END LOOP;  
    RETURN result;  
END conv_integer;
```

```
----- Function call: -----  
...  
y <= conv_integer(a);  
...  
-----
```

This Function converts a STD\_LOGIC\_VECTOR into an INTEGER

#### Function Location:

- Usually placed in a PACKAGE.
- Also possible within main code: Inside ARCHITECTURE or ENTITY
- When used in a PACKAGE, then PACKAGE BODY is necessary, containing the body of each FUNCTION, declared in the declarative part of PACKAGE.





## FUNCTIONS and PROCEDURES

### FUNCTION Example

```
1 -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 ENTITY dff IS
6     PORT ( d, clk, rst: IN STD_LOGIC;
7           q: OUT STD_LOGIC);
8 END dff;
9 -----
10 ARCHITECTURE my_arch OF dff IS
11 -----
12     FUNCTION positive_edge(SIGNAL s: STD_LOGIC)
13     RETURN BOOLEAN IS
14     BEGIN
15         RETURN s'EVENT AND s='1';
16     END positive_edge;
17 -----
18 BEGIN
19     PROCESS (clk, rst)
20     BEGIN
21         IF (rst='1') THEN q <= '0';
22         ELSIF positive_edge(clk) THEN q <= d;
23         END IF;
24     END PROCESS;
25 END my_arch;
26 -----
```

The “positive\_edge” function is placed in the declarative part of ARCHITECTURE (main code) and for creating a DFF within the ARCHITECTURE.

```
1 ----- Package: -----
2 LIBRARY ieee;
3 USE ieee.std_logic_1164.all;
4 -----
5 PACKAGE my_package IS
6     FUNCTION positive_edge(SIGNAL s: STD_LOGIC) RETURN BOOLEAN;
7 END my_package;
8 -----
9 PACKAGE BODY my_package IS
10     FUNCTION positive_edge(SIGNAL s: STD_LOGIC)
11     RETURN BOOLEAN IS
12     BEGIN
13         RETURN s'EVENT AND s='1';
14     END positive_edge;
15 END my_package;
16 -----
17 ----- Main code: -----
18 LIBRARY ieee;
19 USE ieee.std_logic_1164.all;
20 USE work.my_package.all;
21 -----
22 ENTITY dff IS
23     PORT ( d, clk, rst: IN STD_LOGIC;
24           q: OUT STD_LOGIC);
25 END dff;
26 -----
27 ARCHITECTURE my_arch OF dff IS
28 BEGIN
29     PROCESS (clk, rst)
30     BEGIN
31         IF (rst='1') THEN q <= '0';
32         ELSIF positive_edge(clk) THEN q <= d;
33         END IF;
34     END PROCESS;
35 END my_arch;
36 -----
```

The “positive\_edge” function is:

- Placed in a PACKAGE
- It can be reused and shared
- FUNCTION declared in PACKAGE.
- FUNCTION described in PACKAGE BODY

These two codes can be compiled as two separate files or compiled as a single file saved as dff.vhd (entity name).

# FUNCTIONS and PROCEDURES

## PROCEDURE Definition

- PROCEDURE is very similar to FUNCTION and has the same purpose.
- However, a PROCEDURE can **RETURN more than one value**.
- Two parts are required to construct a PROCEDURE: PROCEDURE Body and CALL

### PROCEDURE BODY:

```
PROCEDURE procedure_name [<parameter list>] IS  
    [declarations]  
BEGIN  
    (sequential statements)  
END procedure_name;
```

```
PROCEDURE my_procedure ( a: IN BIT; SIGNAL b, c: IN BIT;  
                        SIGNAL x: OUT BIT_VECTOR(7 DOWNT0 0);  
                        SIGNAL y: INOUT INTEGER RANGE 0 TO 99) IS  
BEGIN  
    ...  
END my_procedure;
```

In this example a, b, c are input **a** is CONSTANT of type BIT (ommission of CONSTANT is allowed), **b and c** are SIGNALS of type BIT.  
Signals x and y are the return signals, mode OUT and INOUT type BIT\_VECTOR and INTEGER.

- *<parameter list>* can be CONSTANT, SIGNAL or VARIABLE. (input and output parameters.)
- More than one return value is allowed.
- For inputs CONSTANT is the default parameter type, while for output is VARIABLE.
- Note that :
  - WAIT , SIGNAL declarations and COMPONENTS are not synthesizable either for FUNCTIONS or PROCEDURES.
  - Exceptionally SIGNAL can be declared in PROCEDURE, but then PROCEDURE must be declared in PROCESS.
  - A synthesizable procedure should not infer registers, i.e. no use of WAIT of any other edge detection.

# FUNCTIONS and PROCEDURES

## PROCEDURE Definition

- PROCEDURE is very similar to FUNCTION and has the same purpose.
- However, a PROCEDURE can **RETURN more than one value**.
- Two parts are required to construct a PROCEDURE: PROCEDURE Body and CALL

### PROCEDURE CALL:

```
compute_min_max(in1, in2, in3, out1, out2);  
    -- statement by itself  
  
divide(dividend, divisor, quotient, remainder);  
    -- statement by itself  
  
IF (a>b) THEN compute_min_max(in1, in2, in3, out1, out2);  
    -- procedure call associated to another statement
```

- PROCEDURE call is a statement on its own. Not like FUNCTION that is part of an expression.
- PROCEDURE can appear but itself or associated to a statement (concurrent or sequential).

# FUNCTIONS and PROCEDURES

## PROCEDURE in the Main Code

```

-----
LIBRARY ieee;
USE ieee.std_logic_1164.all;

-----

ENTITY min_max IS
    GENERIC (limit : INTEGER := 255);
    PORT ( ena: IN BIT;
           inp1, inp2: IN INTEGER RANGE 0 TO limit;
           min_out, max_out: OUT INTEGER RANGE 0 TO limit);
END min_max;


-----

ARCHITECTURE my_architecture OF min_max IS
    -----
    PROCEDURE sort (SIGNAL in1, in2: IN INTEGER RANGE 0 TO limit;
                    SIGNAL min, max: OUT INTEGER RANGE 0 TO limit) IS
    BEGIN
        IF (in1 > in2) THEN
            max <= in1;
            min <= in2;
        ELSE
            max <= in2;
            min <= in1;
        END IF;
    END sort;

    -----
BEGIN
    PROCESS (ena)
    BEGIN
        IF (ena='1') THEN sort (inp1, inp2, min_out, max_out);
        END IF;
    END PROCESS;
END my_architecture;
-----

```

- This code uses PROCEDURE called sort.
- PROCEDURE is located in the declarative part of ARCHITECTURE (main code).
- PROCEDURE call is a statement by its own.
- **What does our sort PROCEDURE do?**





# FUNCTIONS and PROCEDURES

## PROCEDURE in PACKAGE

```
----- Package: -----
LIBRARY ieee;
USE ieee.std_logic_1164.all;

-----
PACKAGE my_package IS
    CONSTANT limit: INTEGER := 255;
    PROCEDURE sort (SIGNAL in1, in2: IN INTEGER RANGE 0 TO limit;
        SIGNAL min, max: OUT INTEGER RANGE 0 TO limit);
END my_package;


-----
PACKAGE BODY my_package IS
    PROCEDURE sort (SIGNAL in1, in2: IN INTEGER RANGE 0 TO limit;
        SIGNAL min, max: OUT INTEGER RANGE 0 TO limit) IS
    BEGIN
        IF (in1 > in2) THEN
            max <= in1;
            min <= in2;
        ELSE
            max <= in2;
            min <= in1;
        END IF;
    END sort;
END my_package;
```

```
----- Main code: -----
LIBRARY ieee;
USE ieee.std_logic_1164.all;
USE work.my_package.all;

-----
ENTITY min_max IS
    GENERIC (limit: INTEGER := 255);
    PORT ( ena: IN BIT;
        inp1, inp2: IN INTEGER RANGE 0 TO limit;
        min_out, max_out: OUT INTEGER RANGE 0 TO limit);
END min_max;

-----
ARCHITECTURE my_architecture OF min_max IS
BEGIN
    PROCESS (ena)
    BEGIN
        IF (ena='1') THEN sort (inp1, inp2, min_out, max_out);
        END IF;
    END PROCESS;
END my_architecture;
```

- This code uses PROCEDURE called sort.
- PROCEDURE is in a PACKAGE called my\_package.
- Now this PROCEDURE can be reused and shared.
- PROCEDURE call is a statement by its own.
- **What does our sort PROCEDURE do?**



# FUNCTIONS and PROCEDURES

## Summary

FUNCTION	PROCEDURE	
Zero or more input parameters. They can only be CONSTANTS (default) or SIGNALS (VARIABLES not allowed)	Any number of IN, OUT or INOUT parameters. They can be CONSTANTS, VARIABLES or SIGNALS	Differences
By default, input parameter is CONSTANT	By default, input (IN) parameter is CONSTANT, and output (OUT and INOUT) parameters are VARIABLE.	
Called as part of an expression	Statement by its own	
WAIT and COMPONENTS are not Synthesizable	WAIT and COMPONENTS are not Synthesizable	In common
Usually placed in PACKAGE. PACKAGE BODY is necessary.	Usually placed in PACKAGE. PACKAGE BODY is necessary.	
Less common, can be placed also into ENTITY or ARCHITECTURE (main code)	Less common, can be placed also into ENTITY or ARCHITECTURE (main code)	

# FUNCTIONS and PROCEDURES

## FUNCTION Exercise

Create a function named **mult()** that can multiply two UNSIGNED values, returning the UNSIGNED product. The parameters to the FUNCTION do not need to have the same number of bits and their order TO/DOWNTO can be any.

```
1  ----- Package: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  USE ieee.std_logic_arith.all;
5  -----
6  PACKAGE pack IS
7      FUNCTION mult(a, b: UNSIGNED) RETURN UNSIGNED;
8  END pack;
9  -----
10 PACKAGE BODY pack IS
11     FUNCTION mult(a, b: UNSIGNED) RETURN UNSIGNED IS
12         CONSTANT max: INTEGER := a'LENGTH + b'LENGTH - 1;
13         VARIABLE aa: UNSIGNED(max DOWNTO 0) :=
14             (max DOWNTO a'LENGTH => '0')
15             & a(a'LENGTH-1 DOWNTO 0);
16         VARIABLE prod: UNSIGNED(max DOWNTO 0) := (OTHERS => '0');
17     BEGIN
18         FOR i IN 0 TO a'LENGTH-1 LOOP
19             IF (b(i)='1') THEN prod := prod + aa;
20             END IF;
21             aa := aa(max-1 DOWNTO 0) & '0';
22         END LOOP;
23         RETURN prod;
24     END mult;
25 END pack;
26 -----

1  ----- Main code: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  USE ieee.std_logic_arith.all;
5  USE work.my_package.all;
6  -----
7  ENTITY multiplier IS
8      GENERIC (size: INTEGER := 4);
9      PORT ( a, b: IN UNSIGNED(size-1 DOWNTO 0);
10           y: OUT UNSIGNED(2*size-1 DOWNTO 0));
11 END multiplier;
12 -----
13 ARCHITECTURE behavior OF multiplier IS
14 BEGIN
15     y <= mult(a,b);
16 END behavior;
17 -----
```

## FUNCTIONS and PROCEDURES

### PROCEDURE Exercise

Use the same FUNCTION made before and transform it into a procedure within a PACKAGE. Add a second output to the procedure at your choice.

Previous example:

Use a function named **mult()** that can multiply two UNSIGNED values, returning the UNSIGNED product. The parameters to the FUNCTION do not need to have the same number of bits and their order TO/DOWNT0 can be any.

```
1  ----- Package: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  USE ieee.std_logic_arith.all;
5  -----
6  PACKAGE pack IS
7      FUNCTION mult(a, b: UNSIGNED) RETURN UNSIGNED;
8  END pack;
9  -----
10 PACKAGE BODY pack IS
11     FUNCTION mult(a, b: UNSIGNED) RETURN UNSIGNED IS
12         CONSTANT max: INTEGER := a'LENGTH + b'LENGTH - 1;
13         VARIABLE aa: UNSIGNED(max DOWNT0 0) :=
14             (max DOWNT0 a'LENGTH => '0')
15             & a(a'LENGTH-1 DOWNT0 0);
16         VARIABLE prod: UNSIGNED(max DOWNT0 0) := (OTHERS => '0');
17     BEGIN
18         FOR i IN 0 TO a'LENGTH-1 LOOP
19             IF (b(i)='1') THEN prod := prod + aa;
20             END IF;
21             aa := aa(max-1 DOWNT0 0) & '0';
22         END LOOP;
23         RETURN prod;
24     END mult;
25 END pack;
26 -----


1  ----- Main code: -----
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  USE ieee.std_logic_arith.all;
5  USE work.my_package.all;
6  -----
7  ENTITY multiplier IS
8      GENERIC (size: INTEGER := 4);
9      PORT ( a, b: IN UNSIGNED(size-1 DOWNT0 0);
10           y: OUT UNSIGNED(2*size-1 DOWNT0 0));
11 END multiplier;
12 -----
13 ARCHITECTURE behavior OF multiplier IS
14 BEGIN
15     y <= mult(a,b);
16 END behavior;
17 -----
```



## Exercise

### Preparation for Project

Develop a Finite Impulse Response Filter (FIR) using the last four entries of the signal. Consider generic coefficients of 8-bits, input of 8bit and output of 10bits, as the figure :



Note: When you multiply two numbers of N-bit and M-bit the output dynamic of the multiplication result is (N+M)-bits. When you perform addition, the number of bit of the result will be incremented by 1.

```

29 library IEEE;
30 use IEEE.std_logic_1164.all;
31 library ieee;
32 use ieee.std_logic_1164.all;
33 use ieee.numeric_std.all;
34 entity fir_filter_4 is
35 port (
36     i_clk      : in  std_logic;
37     i_rstb     : in  std_logic;
38     -- coefficient
39     i_coeff_0   : in  std_logic_vector( 7 downto 0);
40     i_coeff_1   : in  std_logic_vector( 7 downto 0);
41     i_coeff_2   : in  std_logic_vector( 7 downto 0);
42     i_coeff_3   : in  std_logic_vector( 7 downto 0);
43     -- data input
44     i_data      : in  std_logic_vector( 7 downto 0);
45     -- filtered data
46     o_data      : out std_logic_vector( 9 downto 0);
47 end fir_filter_4;
48
49 architecture rtl of fir_filter_4 is
50 type t_data_pipe is array (0 to 3) of signed(7 downto 0);
51 type t_coeff is array (0 to 3) of signed(7 downto 0);
52 type t_mult is array (0 to 3) of signed(15 downto 0);
53 type t_add_st0 is array (0 to 1) of signed(15+1 downto 0);
54 signal r_coeff : t_coeff;
55 signal p_data : t_data_pipe;
56 signal r_mult : t_mult;
57 signal r_add_st0 : t_add_st0;
58 signal r_add_st1 : signed(15+2 downto 0);
59 begin
60 p_input : process (i_rstb,i_clk)
61 begin
62 if(i_rstb='0') then
63     p_data <= (others=>(others=>'0'));
64     r_coeff <= (others=>(others=>'0'));
65 elsif(rising_edge(i_clk)) then
66     p_data <= signed(i_data)&p_data(0 to p_data'length-2);
67     r_coeff(0) <= signed(i_coeff_0);
68     r_coeff(1) <= signed(i_coeff_1);
69     r_coeff(2) <= signed(i_coeff_2);
70     r_coeff(3) <= signed(i_coeff_3);
71 end if;
72 end process p_input;
73 p_mult : process (i_rstb,i_clk)

```

```

74 begin
75 if(i_rstb='0') then
76     r_mult <= (others=>(others=>'0'));
77 elsif(rising_edge(i_clk)) then
78     for k in 0 to 3 loop
79         r_mult(k) <= p_data(k) * r_coeff(k);
80     end loop;
81 end if;
82 end process p_mult;
83 p_add_st0 : process (i_rstb,i_clk)
84 begin
85 if(i_rstb='0') then
86     r_add_st0 <= (others=>(others=>'0'));
87 elsif(rising_edge(i_clk)) then
88     for k in 0 to 1 loop
89         r_add_st0(k) <= resize(r_mult(2*k),17) + resize(r_mult(2*k+1),17);
90     end loop;
91 end if;
92 end process p_add_st0;
93 p_add_st1 : process (i_rstb,i_clk)
94 begin
95 if(i_rstb='0') then
96     r_add_st1 <= (others=>'0');
97 elsif(rising_edge(i_clk)) then
98     r_add_st1 <= resize(r_add_st0(0),18) + resize(r_add_st0(1),18);
99 end if;
100 end process p_add_st1;
101 p_output : process (i_rstb,i_clk)
102 begin
103 if(i_rstb='0') then
104     o_data <= (others=>'0');
105 elsif(rising_edge(i_clk)) then
106     o_data <= std_logic_vector(r_add_st1(17 downto 8));
107 end if;
108 end process p_output;
109 end rtl;

```

Notes:  
[numeric library](#)

# END-FUNCTIONS and PROCEDURES SYSTEM DESIGN

*It is time to integrate several circuits into a bigger system*

**VHDL**  
**VHSIC HARDWARE**  
**DESCRIPTION LANGUAGE**